code:
init 2 python:
    def surprise_interview_requirement():
        if mc.business.is_open_for_business and mc.is_at_work and mc.business.employee_count < mc.business.max_employee_count:
            if corporal_punishment.is_active:
                return True
        return False

    def surprise_teamwork_person(temp_list):
        if any(x for x in temp_list if not x.has_taboo(["sucking_cock", "condomless_sex"]) and x.opinion.threesomes > -2):
            return get_random_from_list([x for x in temp_list if not x.has_taboo(["sucking_cock", "condomless_sex"]) and x.opinion.threesomes > -2])
        return None

    surprise_interview_action = ActionMod("Surprise Interview", surprise_interview_requirement, "surprise_interview_label",
        menu_tooltip = "A woman shows up at the office looking for a job.", category = "Business", is_crisis = True)

label surprise_interview_label():
    $ scene_manager = Scene()
    $ the_person = interview_build_candidates_list(1)[0]
    $ the_person.set_mc_title()
    $ the_person.set_title()
    $ the_person.set_possessive_title()
    $ the_person.generate_home()
    $ the_person.change_job(unemployed_job)
    $ office.add_person(the_person)
    $ the_other_person = mc.business.hr_director
    $ threesome_person = None
    if not the_other_person:
        $ the_other_person = get_random_from_list(mc.business.hr_team)
    if not the_other_person:
        $ the_other_person = get_random_from_list(mc.business.supply_team)
    $ mc.change_location(ceo_office)
    "You are gathering some papers from the desk in your office when you hear an urgent knocking on the door."
    if the_other_person:
        "As you are turning to answer the door you hear someone in the main office start talking."
        the_other_person "Excuse me? Can I help you?"
        "The person at your door responds, but you can't quite make out her words."
        if mc.business.hr_director or get_random_from_list(mc.business.hr_team):
            the_other_person "I'm sorry, I didn't know he was expecting someone, let me check the calendar. What was your name?"
            "Her reply is indistinct again, but it sounds a bit more flustered, almost panicky."
        else:
            the_other_person "Oh, sorry we don't have anyone handling human resources right now, I'm sure he'll be right out."
    else:
        "There isn't anyone in the outer office, so you should probably see who is there."
    $ mc.change_location(office)
    $ scene_manager.add_actor(the_person)
    $ scene_manager.update_actor(the_person, position = "walking_away")
    "You make your way to the door and open it to see a stranger standing just outside."
    if the_other_person:
        $ scene_manager.add_actor(the_other_person, display_transform = character_center_flipped, emotion = "sad")
        "Right behind her is [the_other_person.possessive_title], obviously trying to figure out what to do with this person just walking into the office."
    $ scene_manager.update_actor(the_person, position = "stand2", emotion = "happy")
    the_person "Oh, hello Mr. [mc.last_name]! I'm [the_person.fname]. I'm here for my interview."
    "Interview? Did you have an interview scheduled today?"
    if get_random_from_list(mc.business.hr_team):
        the_other_person "I'm sorry [the_other_person.mc_title] I didn't see anything about an interview. Did you schedule one?"
        mc.name "I don't think so... but since she is here, can you pull up her application?"
        the_other_person "Of course."
        $ scene_manager.update_actor(the_other_person, display_transform = character_center_flipped, position = "standing_doggy")
    elif get_random_from_list(mc.business.supply_team):
        "[the_other_person.possessive_title!c] looks relieved to have you taking care of this and gets back to her work ordering supplies."
        $ scene_manager.update_actor(the_other_person, display_transform = character_center_flipped, position = "sitting")
    else:
        "At least no one else is here to see you messing up a simple interview schedule."
    mc.name "I'm sorry, [the_person.fname] was it? Did I set up this interview?"
    the_person "Not personally, I got an email from one of your employees."
    if not get_random_from_list(mc.business.hr_team):
        "None of your employees are even responsible for hiring..."
    mc.name "Sorry, for the confusion, could I see the email? I want to make sure we follow up so we can get you an apology from the person responsible."
    the_person "Oh... right... it should be right here."
    "She starts looking through her phone, but it is pretty clear she isn't finding the email. In fact, it seems like she doesn't even expect to find one."
    if get_random_from_list(mc.business.hr_team):
        $ scene_manager.update_actor(the_other_person, display_transform = character_center_flipped, emotion = "sad")
        "As she gets more and more flustered [the_other_person.title] walks up empty-handed."
        the_other_person "I'm sorry [the_other_person.mc_title] I can't find her application."
        "She hasn't even submitted an application?"
    mc.name "Do you at least remember the name of the person who emailed you?"
    $ scene_manager.update_actor(the_person, position = "kneeling1", emotion = "sad")
    "That seems to break her, she drops to her knees and seems to be practically in tears."
    the_person "Oh, this was so stupid. I'm so sorry! I don't have an interview, I've just been so frustrated with my job hunting."
    if the_person.age < 35:
        the_person "I just keep sending in applications... no one even responds... my father said... back in his day..."
    else:
        the_person "I just keep sending in applications... no one even responds... my grandfather said... back in his day..."
    the_person "I hoped... I don't know... I should just go..."
    if get_random_from_list(mc.business.hr_team):
        $ scene_manager.update_actor(the_other_person, display_transform = character_center_flipped, emotion = "angry")
        "It looks like [the_other_person.title] is about to read her the riot act."
    elif get_random_from_list(mc.business.supply_team):
        $ scene_manager.update_actor(the_other_person, display_transform = character_center_flipped, emotion = "happy", position = "sitting")
        "Out of the corner of your eye you catch [the_other_person.title] looking up from her work, struggling to contain a fit of giggles."
    menu:
        "Have mercy on her":
            mc.name "That's not necessary. Why don't you have a seat on the couch in my office and collect yourself? I'll give you a few minutes and then we can start over."
            the_person "*sniff* R-really?"
            mc.name "Yes, we can straighten this out, please stop crying."
            the_person "Oh, Thank you so much!"
            "She quickly rushes into your office, possibly a little concerned you are going to change your mind."
            $ scene_manager.remove_actor(the_person)
            if get_random_from_list(mc.business.hr_team):
                "[the_other_person.possessive_title!c] gives the door time to shut but it is clear she is about to turn her anger on you. You hold up your hands to forestall the tirade."
                mc.name "I know, I know, but clearly she is having a hard time. I'm just gonna talk to her a bit. Who knows she might be a good employee."
                the_other_person "Whatever, it is your business."
                "[the_other_person.title] rolls her eyes and turns to get back to her actual work."
                $ scene_manager.remove_actor(the_other_person)
            elif get_random_from_list(mc.business.supply_team):
                "You give [the_other_person.title] a glare and she quickly ducks her head back to her work."
                $ scene_manager.remove_actor(the_other_person)
            "To pass the time you walk over to the coffee machine and fill a couple of cups."
            call give_serum(the_person) from _call_give_serum_surprise
        "Kick her out":
            mc.name "I think that would be for the best. You've wasted enough of our time."
            mc.name "Do I need to escort you out of the building or do you think you can make it out properly?"
            $ scene_manager.update_actor(the_person, position = "stand2", emotion = "sad")
            "She manages to pull herself together and stand up, wiping at her eyes."
            the_person "No, I'll be fine. Sorry again."
            $ scene_manager.update_actor(the_person, position = "walking_away", emotion = "sad")
            if get_random_from_list(mc.business.hr_team):
                "She turns to go and [the_other_person.title] turns to look at you with an incredulous look on her face."
                "At a bit of a loss you just shrug your shoulders and head back into your office."
            elif get_random_from_list(mc.business.supply_team):
                "She turns to go and thankfully makes it out of the room before [the_other_person.title] breaks into full out laughter."
                "At a bit of a loss you just shrug your shoulders and head back into your office."
            else:
                "She turns to go and you watch to make sure she leaves the office, debating on if you should head down to the lobby to ensure she is gone."
            $ scene_manager.remove_actor(the_person)
            $ the_person.remove_person_from_game()
            $ scene_manager.clear_scene()
            $ mc.change_location(ceo_office)
            return
    $ the_person.add_situational_obedience("bribe", 30, "I need a job!")
    $ mc.change_location(ceo_office)
    $ scene_manager.add_actor(the_person, position = "sitting", emotion = "happy")
    "When you enter your office you find her waiting patiently on the couch as instructed. You walk over and offer her the coffee."
    the_person "Thank you so much, I really messed this up and I'm so sorry."
    mc.name "It's okay, let's start over. I'm going to need to take a look at your application before we proceed."
    the_person "Right of course. I have it right here."
    "She hands you a folder of papers, which you look over while she starts to work on her coffee."
    call hire_select_process([the_person, 1]) from _call_hire_select_process_surprise
    $ scene_manager.add_actor(the_person, position = "sitting", emotion = "happy")
    if not isinstance(_return, Person):
        "After a quick read it is clear you don't actually want [the_person.fname] to work for you, but there is no reason to tell her that right away."
        "While you read she drank most of her coffee and should be ready for some questions."
    else:
        "You've read most of the relevant information from her resume, and she has had time to drink most of her coffee. Now would be a good time to see how well she will fit in with your company."
    mc.name "This looks good, but there are some things you need to know about working here before we go any further."
    the_person "Of course!"
    if corporal_punishment.is_active:
        mc.name "To ensure that everyone is completing their tasks to the best of their ability we have a system of punishments for those who fail."
        the_person "What kind of punishments?"
        mc.name "It varies, they start out small, but can escalate depending on the situation."
        mc.name "You could be assigned extra tasks, or your pay could be cut to make it up to the company."
        mc.name "However, one of the things that I've found to work best is corporal punishment."
        if strict_enforcement.is_active:
            mc.name "In some extreme situations we make an example of the employee who is being punished."
            mc.name "That can mean giving them embarrassing work to do."
            if strict_uniform_policy.is_active:
                mc.name "It can also mean forcing them to wear a uniform that is more shocking than they are used to."
            if draconian_enforcement.is_active:
                mc.name "If all else fails we may force the employee to sacrifice their time and their body to help the company recover."
        the_person "That is pretty robust, but I only have to worry about it if I do something wrong, right?"
        mc.name "Yes, but that brings me to the fact that you have already done something wrong."
        the_person "But I didn't know about these policies, and I don't even work here yet."
        mc.name "That is correct, but if you want to I'm going to need you to demonstrate your commitment."
        mc.name "We'll start out easy, a little spanking to drive the lesson home and then we can proceed."
        mc.name "Turn around and put your hands on the desk."
        if the_person.obedience + 10*the_person.opinion.being_submissive < 120: #She complains
            "She hesitates, then sighs and turns around and plants her hands flat on the desk."
            the_person "Fine..."
        else:
            "[the_person.title] follows your instructions without any hesitation."
        $ mc.change_locked_clarity(10)
        $ scene_manager.update_actor(the_person, position = "standing_doggy")
        "You stand to the side of [the_person.possessive_title] and place one hand on her hip, ready to spank her with the other."
        $ top_clothing = the_person.outfit.get_lower_top_layer
        if top_clothing and top_clothing.can_be_half_off and top_clothing.half_off_gives_access and top_clothing.hide_below and not top_clothing.anchor_below and not top_clothing.underwear:
            menu:
                "Pull up her [top_clothing.display_name] first":
                    if the_person.obedience + 10*the_person.opinion.being_submissive < 120 and the_person.effective_sluttiness("underwear_nudity") < 30:
                        "You grab the hem of [the_person.title]'s [top_clothing.display_name]."
                        the_person "Hey! What are you doing?"
                        mc.name "Making sure your [top_clothing.display_name] doesn't get in the way of your punishment."
                        the_person "Are... Are you allowed to do that?"
                        if reduced_coverage_uniform_policy.is_active:
                            mc.name "Once you work here, I have authority over everything you wear."
                        else:
                            mc.name "No clothing is being removed, so yes I can."
                        "You pull her [top_clothing.display_name] up and leave it bunched around her waist."
                    else:
                        "You grab the hem of [the_person.title]'s [top_clothing.display_name] and pull it up around her waist."
                    $ mc.change_locked_clarity(10)
                    $ the_person.draw_animated_removal(top_clothing, position = "standing_doggy", half_off_instead = True)
                    if not the_person.wearing_panties:
                        mc.name "No panties today, I see."
                    $ the_person.update_outfit_taboos()
                "Leave her [top_clothing.display_name] in place":
                    pass
        "You rub her butt briefly, then slap it hard."
        $ not_cushioned = the_person.vagina_visible or (the_person.wearing_panties and not the_person.outfit.panties_covered)
        if not_cushioned: #ouch!
            $ the_person.slap_ass()
            "Your hand makes a satisfying smack as it makes contact with her ass cheek. Her ass jiggles for a few moments before settling down."
            the_person "Ah... That really stings..."
        else: #Not too bad.
            "Her clothing absorbs some of the blow, but you still make good contact and set her ass jiggling for a moment."
            the_person "Ah..."
        "You keep smacking her butt, putting more force behind your blow each time."
        if not_cushioned: #Ass gets red, she gets sore.
            $ mc.change_locked_clarity(10)
            "Her exposed ass jiggles with each hit, and quickly starts to turn red."
            the_person "Ah... Am I almost done [the_person.mc_title]?"
            $ the_person.slap_ass()
            if the_person.opinion.being_submissive > 0: #She likes it and is getting turned on.
                "You spank her again, and she moans."
                $ the_person.discover_opinion("being submissive")
                $ the_person.change_arousal(5*the_person.opinion.being_submissive)
                the_person "I... Don't know if I can take much more of this! Mmph..."
            else:
                "You spank her again, making her gasp."
                the_person "I... Don't know how much more of this I can take!"
            mc.name "You'll take it until I think you have learned your lesson. Do you understand?"
            "Another smack, another ass jiggle."
            if the_person.opinion.being_submissive > 0:
                the_person "Yes [the_person.mc_title]! I understand! Ah!"
            else:
                "She lowers her head and grits her teeth."
                the_person "Yes [the_person.mc_title]. Ah..."
            mc.name "Do you have anything to say for your actions?"
            if the_person.opinion.being_submissive > 0:
                the_person "It won't happen again [the_person.mc_title], I..."
                "You interrupt her with a slap on the ass. She pauses, then continues."
                the_person "I'm sorry, and I see just how wrong I was now! Ah!"
            else:
                the_person "Ow... It won't happen again [the_person.mc_title], I... Ow!"
                "You interrupt her with a slap on the ass. She takes a moment to collect herself before continuing."
                the_person "I promise it won't happen again!"
            $ mc.change_locked_clarity(10)
            "You give her one last hit on her now red butt and then step back, letting her stand up."
            $ the_person.slap_ass()
            $ the_person.outfit.restore_all_clothing()
            $ scene_manager.update_actor(the_person, position = "stand2")
            if the_person.opinion.being_submissive > 0:
                $ the_person.change_arousal(5*the_person.opinion.being_submissive)
                $ the_person.change_obedience(the_person.opinion.being_submissive)
                the_person "Thank you [the_person.mc_title]. I promise I'll do better."
            else:
                $ the_person.change_obedience(3)
                $ the_person.change_love(-2 + the_person.opinion.being_submissive)
                the_person "Finally. Ow..."
            mc.name "I expect you've learned your lesson. Would you like to continue?"
            the_person "Yes, I still want to work here."
        else: #Doesn't particularly mind."
            if top_clothing:
                "[the_person.title] jerks forward with each strike, but her [top_clothing.display_name] seems to be saving her from the worst of it."
            else:
                "[the_person.title] jerks forward with each strike, but she doesn't seem to mind getting spanked."
            the_person "Ah... Ow..."
            "After a few more strikes it's clear you aren't having the effect on [the_person.possessive_title] that you were hoping for."
            "You give her one last slap on the ass and step back."
            mc.name "Stand up, we're done here."
            $ the_person.outfit.restore_all_clothing()
            $ scene_manager.update_actor(the_person, position = "stand2")
            $ mc.change_locked_clarity(10)
            "She turns around, rubbing her butt."
            the_person "Can we continue the interview?"
            $ the_person.change_love(-1)
            $ the_person.change_obedience(2)
    if office_conduct_guidelines.is_active:
        mc.name "Continuing on with our expectations, there are a number of conduct guidelines we have for employees that I'd like you to read over."
        "You hand her the guidelines your staff have been reading along with the five latest daily emails."
        if the_person.sluttiness <20:
            "As she reads them over, she seems a bit flustered, but she doesn't object to anything that is there."
            $ the_person.change_slut(5, 20)
        else:
            "As she reads them over she doesn't seem surprised or concerned by any of the contents."
        if mandatory_staff_reading.is_active:
            mc.name "For further reading I also have this book for you."
            "You pull out a copy of \"Your Place in the Work Place\" and flip it to an especially important part."
            mc.name "I obviously don't expect you to read it all now, but this passage is one of my favourites."
            if the_person.sluttiness <20:
                $ the_person.change_happiness(-5)
            if the_person.sluttiness < 40:
                $ the_person.change_slut(5, 40)
                "As she skims the page you see some significant blushing start to colour her skin."
                the_person "Wow, that is rather detailed."
            else:
                "She is still unfazed by the growing sexuality of these materials."
                the_person "Wow, this book is pretty good."
            if superliminal_office_messaging.is_active:
                mc.name "I also have a company calendar and some posters for you. If you are hired you can put them in your work area or take them home."
                if the_person.sluttiness <20:
                    "She smiles at first, but when she sees they are all of nude women it falls into something approaching a frown."
                    $ scene_manager.update_actor(the_person, emotion = "sad")
                    $ the_person.change_happiness(-5)
                else:
                    "She smiles at first, it slips a bit when she notices they all feature nude women, but she manages to keep it forced in place anyway."
                $ the_person.change_slut(5, 60)
    if strict_uniform_policy.is_active:
        mc.name "You will be required to wear a uniform for work, fortunately the company will provide it for you."
        the_person "That is so generous, is there a special store I need to go to?"
        mc.name "Actually, I'm pretty sure I have one your size here. Let me check the closet."
        "As you walk over to search you continue talking over your shoulder."
        mc.name "We have a very robust uniform policy here, the clothing you are provided is exactly what you need to wear."
        if maximal_arousal_uniform_policy.is_active:
            mc.name "It can be difficult to get used to at first, but most girls get comfortable with them eventually."
            $ slut_limit = 999 #ie. no limit at all.
        elif corporate_enforced_nudity_policy.is_active:
            mc.name "It can be difficult to get used to at first, but most girls adapt over time."
            $ slut_limit = 80
        elif minimal_coverage_uniform_policy.is_active:
            mc.name "It can be difficult to get used to at first, but everyone adjusts pretty quickly."
            $ slut_limit = 60
        elif reduced_coverage_uniform_policy.is_active:
            mc.name "It can be a bit unusual, but nothing too crazy."
            $ slut_limit = 40
        elif casual_uniform_policy.is_active:
            mc.name "It is pretty basic, we just want to highlight our staff."
            $ slut_limit = 25
        elif relaxed_uniform_policy.is_active:
            mc.name "It is pretty basic, we just want to highlight our staff."
            $ slut_limit = 15
        else:
            mc.name "It is pretty basic, we just want to highlight our staff."
            $ slut_limit = 5
        call outfit_master_manager(slut_limit = slut_limit, show_outfits = False, show_underwear = False, show_overwear = True) from _call_outfit_master_manager_surprise
        $ the_outfit = _return
        if not the_outfit:
            $ the_outfit = default_wardrobe.get_random_appropriate_overwear(slut_limit, sluttiness_min = slut_limit-20, guarantee_output = True, preferences = WardrobePreference(the_person))
        mc.name "Ah, yes here it is."
        "You turn with the outfit in hand and look at [the_person.title]."
        mc.name "Let's start by having you strip down."
        if the_person.effective_sluttiness("underwear_nudity") + (the_person.opinion.not_wearing_anything*10) < 30:
            the_person "You mean right here?"
            mc.name "Yes, if you are going to work here you need to be comfortable working around me, and this is a great way to see how comfortable you are."
        else:
            the_person "Of course."
        $ generalised_strip_description(the_person, strip_list = the_person.outfit.get_underwear_strip_list())
        $ scene_manager.update_actor(the_person, position = "stand2")
        if the_person.effective_sluttiness("underwear_nudity") + (the_person.opinion.not_wearing_anything*10) < 30:
            "[the_person.possessive_title!c] blushes and tries to cover her body."
            the_person "This is so embarrassing..."
            mc.name "I'm sorry to hear that but..."
        else:
            "[the_person.possessive_title!c] doesn't seem too bothered, but she does use her hands to cover herself slightly."
            the_person "Can I have my uniform now?"
        if commando_uniform_policy.is_active:
            mc.name "You aren't done yet."
            mc.name "The company uniform is every piece of clothing you will wear to work, and at this time that will never include underwear of any kind."
            if the_person.effective_sluttiness(["bare_tits", "bare_pussy"]) + (the_person.opinion.not_wearing_anything*10) < 50:
                the_person "You can't be serious?"
                mc.name "Absolutely! Take a walk through the building and you'll see that it is something everyone is already complying with."
            else:
                the_person "...okay."
                mc.name "You'll see, everyone else in the building is already complying."
            if the_other_person:
                the_person "I mean... I saw [the_other_person.fname], but I thought she was just some kind of exhibitionist freak."
                mc.name "Have you already forgotten our talk about punishments?"
                the_person "Sorry, no sir."
                mc.name "Good, now hurry up."
            else:
                the_person "Now that you mention it, on the way in I saw some rather unique outfits on display."
                mc.name "Exactly, now hurry up."
            $ generalised_strip_description(the_person, strip_list = the_person.outfit.get_full_strip_list())
            $ scene_manager.update_actor(the_person, position = "stand2")
            if the_person.update_outfit_taboos(): # Broke a taboo
                "[the_person.title] stands meekly in front of you, completely nude. She tries to cover herself up with her hands."
        else:
            if the_person.outfit.get_bra():
                $ the_outfit.add_upper(the_person.outfit.get_bra())
            if the_person.outfit.get_panties():
                $ the_outfit.add_lower(the_person.outfit.get_panties())
        mc.name "Hands down, there's no point hiding anything from me now."
        "She frowns, but follows your instructions. She lowers her hands to her sides, letting you get a good view of her body."
        "Satisfied with your view you hand over the uniform you picked out and watch as she starts to put it on."
        if mandatory_vibe_policy.is_active:
            mc.name "Oh, wait, I forgot there is one more part to your uniform I need to give you."
            "You reach into your desk drawer and pull out a bullet vibrator then take a step towards her."
            if the_person.opinion.masturbating <=0:
                the_person "What is that, some kind of pass key?"
                mc.name "It is required when you enter the building, but it doesn't actually unlock the doors."
                mc.name "This is a vibrator, and I'm going to need to insert it into your pussy."
            else:
                the_person "A vibrator?"
                mc.name "Yes, I'm going to need to insert it into your pussy."
            if the_person.effective_sluttiness("touching_vagina") < 55:
                the_person "I think I might need to draw the line there. This is all just too much."
                mc.name "You've come so far [the_person.title]. We are nearly done. Please."
                the_person "...okay, just do it."
            else:
                the_person "If it's really necessary."
                mc.name "Don't worry, I'll be gentle."
            "You step forward and reach out your hand resting the vibe on her lower belly."
            if not commando_uniform_policy.is_active:
                "Slowly you make your way down her body, sliding under the waist of her panties and pressing the toy firmly against her lips."
            else:
                "Slowly you make your way down her body, pressing the toy firmly against her lips."
            if the_person.effective_sluttiness("touching_vagina") < 55:
                "She shivers a bit as you start to push it in but doesn't object further."
            $ the_person.break_taboo("touching_vagina")
            "With a firm push you slip it into her and stroke her pussy as you pull your hand back."
            "Once you are done you flip the switch in your other hand and she lets out a faint moan."
        mc.name "Great, now you can get dressed."
        $ the_person.change_arousal(20)
        $ scene_manager.apply_outfit(the_person, the_outfit)
        $ scene_manager.update_actor(the_person, position = "stand2")
        if the_person.effective_sluttiness() < the_outfit.outfit_slut_score:
            the_person "Is this it? This is so embarrassing..."
        else:
            the_person "I guess this isn't so bad now that I'm dressed again."
        $ the_person.event_triggers_dict["forced_uniform"] = the_outfit.get_copy()
        $ slut_change = 0
        if the_person.tits_visible:
            $ slut_change += the_person.opinion.showing_her_tits
        if the_person.vagina_visible:
            $ slut_change += the_person.opinion.showing_her_ass
        if the_person.tits_visible and the_person.vagina_visible:
            $ slut_change += the_person.opinion.not_wearing_anything
        $ the_person.change_slut(slut_change, 30)
    if the_other_person and surprise_teamwork_person(mc.business.employee_list):
        mc.name "Just one last thing, here at [mc.business.name] we are all part of a team, and it is important that you can work with the team to achieve shared goals."
        mc.name "Your first impression has left something to be desired, so I'm going to call in someone to ensure you can fit in here."
        if surprise_teamwork_person([the_other_person]):
            $ threesome_person = surprise_teamwork_person([the_other_person])
            mc.name "Since [threesome_person.fname] saw you making a scene she is going to be the best person for this."
        elif surprise_teamwork_person(mc.business.hr_team):
            $ threesome_person = surprise_teamwork_person(mc.business.hr_team)
            if the_other_person in mc.business.hr_team:
                mc.name "Since [the_other_person.fname] saw you making a scene we are going to get someone less biased from HR to conduct this test."
            else:
                mc.name "[threesome_person.fname] from our HR department is going to be the best person to conduct this test."
        elif surprise_teamwork_person(mc.business.employee_list):
            $ threesome_person = surprise_teamwork_person(mc.business.employee_list)
            mc.name "Since [the_other_person.fname] saw you making a scene we are going to get someone less biased to conduct this test."
        if threesome_person in mc.business.hr_team or threesome_person in mc.business.supply_team:
            "You walk to the door and call out to [threesome_person.title]."
        else:
            $ temp_dept = threesome_person.get_destination().formal_name
            "You pick up your phone and call down to the [temp_dept] deaprtment to get [threesome_person.title]."
        mc.name "She'll be right here."
        $ scene_manager.add_actor(threesome_person, display_transform = character_center_flipped)
        "Sure enough, [threesome_person.possessive_title] soon walks into your office. She spares a glance for [the_person.title] before turning to you."
        threesome_person "What can I do for you [threesome_person.mc_title]?"
        if threesome_person != the_other_person:
            mc.name "This is [the_person.fname], I was just telling her how important teamwork is and was hoping you could aid in a demonstration."
        else:
            mc.name "I was just telling [the_person.fname] how important teamwork is and was hoping you could aid in a demonstration."
        threesome_person "Of course, what do you need me to do?"
        mc.name "Come over here and get down on your knees."
        if threesome_person.opinion.giving_blowjobs > 0:
            threesome_person "Oh good, I love team building exercises!"
        else:
            threesome_person "Yes [threesome_person.mc_title]."
        "[threesome_person.possessive_title!c] quickly complies with your command, getting herself into position on the floor in front of you."
        $ scene_manager.update_actor(threesome_person, display_transform = character_center_flipped, position = "kneeling1")
        mc.name "You too [the_person.fname]."
        if the_person.effective_sluttiness("sucking_cock") + 5*the_person.opinion.giving_blowjobs < 40:
            the_person "Wait, seriously?"
            threesome_person "Don't worry, I'll help. Most of the girls here love helping out Mr. [l_name]."
            $ scene_manager.update_actor(threesome_person, display_transform = character_center_flipped, position = "blowjob", emotion = "happy", special_modifier = "blowjob")
            "Obviously not feeling the need to wait [threesome_person.title] pulls open your pants and quickly takes you into her mouth."
            "[the_person.title] averts her eyes at first, but as [threesome_person.title] moans around your rapidly hardening cock her eyes start to drift back."
            "For a couple minutes the room is quiet apart from a soft slurping sound each time [threesome_person.title] slides herself down your shaft."
            if mandatory_vibe_policy.is_active:
                $ the_person.change_arousal(10)
                "You goose the vibrator once more and [the_person.title] lets out another moan as one hand drifts to a breast the other drops to her crotch."
            else:
                "At one point you look up to see that [the_person.title]'s hands have drifted, one to her mouth and one to a breast."
            "[the_person.title] seems entranced. As you watch she starts to slowly move forward, taking small steps until she is right next to you."
            $ scene_manager.update_actor(the_person, position = "kneeling1")
            "Another moment later she lowers herself to her knees, staring at [threesome_person.title] as she works over your dick."
            "[threesome_person.title] pulls off and turns to [the_person.title] with a slobbery smile."
            threesome_person "Welcome to the team [the_person.fname], ready to get in the game?"
            "At a loss for words [the_person.title] just nods and leans forward. [threesome_person.title] gives you a quick grin and joins her."
        else:
            the_person "...okay."
            threesome_person "Don't worry, you'll like this. He has such a nice cock and is always happy to share it with us."
            "[the_person.possessive_title!c] also drops to her knees and although she looks less eager she prepares herself as you open up your pants."
        call start_threesome(threesome_person, the_person, start_position = threesome_double_blowjob, skip_intro = True, private = True, position_locked = True, affair_ask_after = False) from _call_start_threesome_surprise
        $ threesome_person.restore_all_clothing()
        $ scene_manager.update_actor(threesome_person, position = "stand2")
        "With her work done, [threesome_person.title] gets up and starts to straighten her clothes."
        $ scene_manager.update_actor(the_person, position = "kneeling1", display_transform = character_right)
        "[the_person.title] is a bit slower, and stays on her knees. It seems she is still shocked by how far things went."
        $ scene_manager.update_actor(threesome_person, position = "walking_away")
        "[threesome_person.title] leans over and gives [the_person.title] a kiss on the top of her head then turns to head for the door."
        if the_other_person == threesome_person:
            $ town_relationships.update_relationship(the_person, threesome_person, "Friend")
            threesome_person "I guess having you work here will be fun after all."
        else:
            $ town_relationships.update_relationship(the_person, threesome_person, "Best Friend")
            threesome_person "I'm really glad you are working here now, we are going to have so much fun together."
        $ scene_manager.remove_actor(threesome_person)
        if strict_uniform_policy.is_active:
            $ scene_manager.apply_outfit(the_person, the_outfit)
        $ scene_manager.update_actor(the_person, position = "stand2", display_transform = character_right)
        "As [threesome_person.title] leaves the office, [the_person.title] finally makes it back to her feet."
        the_person "I've never had an interview like this before. Is it done? Was that all?"
    else:
        mc.name "Great, you've done wonderful and addressed all my concerns."
        the_person "That's all?"
    mc.name "That is all, do you have any questions?"
    the_person "When can I start?"
    "Well, she certainly has some gumption. Do you want to hire her?"
    $ the_person.clear_situational_obedience("bribe")
    menu:
        "Hire her":
            call hire_someone(the_person) from _call_hire_someone_surprise
            $ the_person.change_stats(happiness = 5, love = 2)
            $ the_person.change_location(the_person.get_destination())
            mc.name "Right now. Head down the hall to your new department and get started."
        "Not now":
            $ mc.phone.register_number(the_person)
            $ the_person.change_location(the_person.home)
            mc.name "I'm sorry, but this just isn't going to work out right now."
            mc.name "You did very well but we just aren't looking to expand."
            the_person "What!? I let you... and now you're just going to kick me out anyway?"
            mc.name "I'm very sorry, I have your number and will be sure to call if something opens up in the future."
        "Never":
            mc.name "I'm sorry, but this just isn't going to work out."
            mc.name "You came in here lying and I'd never be able to trust you. This was just a way to get revenge for you wasting my time."
            the_person "Oh my god!? I can't believe I let you... I'm gonna go home and boil my skin off in the shower."
            $ scene_manager.remove_actor(the_person)
            $ the_person.remove_person_from_game()
    $ scene_manager.clear_scene()
    $ clean_memory()
    return


code:
init 2 python:
    def uniform_designer_requirement():
        person = mc.business.event_triggers_dict.get("uniform_designer", None)
        if person:
            if not person.is_employee:
                mc.business.event_triggers_dict["uniform_designer"] = None
        if strict_uniform_policy.is_active:
            if mc.is_at_work:
                if not mc.business.event_triggers_dict.get("uniform_designer", None):
                    if mc.business.employee_count > 3:
                        if mc.business.is_open_for_business:
                            return True
        return False

    def design_others_outfit(designer, model, temp_sluttiness_boost = 0, target = 0):
        if designer.effective_sluttiness() >= model.effective_sluttiness():
            outfit_slut_points = builtins.min(builtins.int((designer.effective_sluttiness() + temp_sluttiness_boost) / 8.0), 12.0)
            allow_skimpy = False
            model.change_stats(slut = 1, max_slut = (30 if designer.effective_sluttiness() >= 30 else designer.effective_sluttiness()))
        else:
            outfit_slut_points = builtins.min(builtins.int((model.effective_sluttiness() + temp_sluttiness_boost) / 8.0), 12.0)
            allow_skimpy = True
        builder = WardrobeBuilder(designer)
        thinks_appropriate = False
        if outfit_slut_points < target/12:
            outfit_slut_points = target/12
        while not (thinks_appropriate or outfit_slut_points < 0):
            outfit = model.personalize_outfit(builder.build_outfit(None, outfit_slut_points), opinion_color = designer.favourite_colour, coloured_underwear = True, swap_bottoms = True, allow_skimpy = allow_skimpy)
            thinks_appropriate = model.judge_outfit(outfit, temp_sluttiness_boost = builtins.int(temp_sluttiness_boost/4.0))
            outfit_slut_points -= 2
        if thinks_appropriate and outfit:
            return outfit
        return None

    def design_test(designer):
        temp_sluttiness_boost = 0
        target = 0
        if strict_uniform_policy.is_active:
            temp_sluttiness_boost +=4
            target = 10
            if reduced_coverage_uniform_policy.is_active:
                temp_sluttiness_boost +=4
                target = 20
                if casual_uniform_policy.is_active:
                    temp_sluttiness_boost +=4
                    target = 30
                    if minimal_coverage_uniform_policy.is_active:
                        temp_sluttiness_boost +=4
                        target = 40
                        if minimal_coverage_uniform_policy.is_active:
                            temp_sluttiness_boost +=4
                            target = 60
                            if corporate_enforced_nudity_policy.is_active:
                                temp_sluttiness_boost +=4
                                target = 80
                                if maximal_arousal_uniform_policy.is_active:
                                    temp_sluttiness_boost +=4
                                    target = 100
        for person in mc.business.employee_list:
            outfit = design_others_outfit(designer, person, temp_sluttiness_boost = temp_sluttiness_boost, target = target)
            if outfit:
                person.apply_outfit(outfit.get_copy())
        return

    def full_staff_design(designer):
        temp_sluttiness_boost = 0
        target = 0
        if strict_uniform_policy.is_active:
            temp_sluttiness_boost +=4
            target = 10
            if reduced_coverage_uniform_policy.is_active:
                temp_sluttiness_boost +=4
                target = 20
                if casual_uniform_policy.is_active:
                    temp_sluttiness_boost +=4
                    target = 30
                    if minimal_coverage_uniform_policy.is_active:
                        temp_sluttiness_boost +=4
                        target = 40
                        if minimal_coverage_uniform_policy.is_active:
                            temp_sluttiness_boost +=4
                            target = 60
                            if corporate_enforced_nudity_policy.is_active:
                                temp_sluttiness_boost +=4
                                target = 80
                                if maximal_arousal_uniform_policy.is_active:
                                    temp_sluttiness_boost +=4
                                    target = 999
        for person in mc.business.employee_list:
            outfit = design_others_outfit(designer, person, temp_sluttiness_boost = temp_sluttiness_boost, target = target)
            if outfit:
                person.event_triggers_dict["forced_uniform"] = outfit.get_copy()
        return

    def full_staff_clear():
        for person in mc.business.employee_list:
            person.event_triggers_dict["forced_uniform"] = None
        return

    def uniform_prep_requirement(): # monday morning
        if day%7 == 0:
            if time_of_day == 0:
                person = mc.business.event_triggers_dict.get("uniform_designer", None)
                if person:
                    if not person.is_employee:
                        person = None
                        mc.business.event_triggers_dict["uniform_designer"] = None
                if person:
                    return True
        return False

    def uniform_followup_requirement():
        if mc.business.event_triggers_dict.get("uniform_designer", None):
            if day%7 == 2 or day%7 == 3:
                if mc.is_at_work:
                    if mc.business.event_triggers_dict.get("uniform_designer", None).is_at_office:
                        if len(build_forced_uniform_list(mc.business.event_triggers_dict.get("uniform_designer", None))) > 2:
                            return True
        return False

    uniform_designer_action = ActionMod("Uniform Designer", uniform_designer_requirement, "uniform_designer_action_label",
        menu_tooltip = "An employee wants to talk about the work uniforms.", category = "Business Crisis", is_crisis = True)

    def get_uniform_designer():
        temp_list = []
        opinion = -2
        slut = 0
        for person in mc.business.employee_list:
            person.create_opinion("work uniforms", start_positive = False, start_known = False, add_to_log = False)
        while not temp_list:
            for person in mc.business.employee_list:
                if person.get_opinion_score("work uniforms") == opinion:
                    temp_list.append(person)
            opinion += 1
        person = get_random_from_list(temp_list)
        for x in temp_list:
            if x.sluttiness > person.sluttiness:
                person = x
        return person

    def build_forced_uniform_list(person):
        temp_list = mc.business.employee_list
        if person in temp_list:
            temp_list.remove(person)
        for other_person in temp_list:
            if not other_person.event_triggers_dict.get("forced_uniform", None):
                temp_list.remove(other_person)
        return temp_list

    def uniform_followup_prep():
        temp_list = mc.business.employee_list
        for person in temp_list:
            if person.event_triggers_dict.get("forced_uniform", None):
                person.apply_outfit(person.event_triggers_dict.get("forced_uniform", None).get_copy())
        return

label uniform_designer_action_label():
    $ scene_manager = Scene()
    $ the_person = get_uniform_designer()
    $ mc.business.event_triggers_dict["uniform_designer"] = the_person
    if the_person is None:
        return
    $ day_part = time_of_day_string(time_of_day)
    $ done = False
    if (mc.location == the_person.get_destination()):
        "You're hard at work in the [day_part], when [the_person.possessive_title] asks you over to discuss something."
    else:
        "You're hard at work in the [day_part], when [the_person.possessive_title] calls you on your phone to discuss something."
    $ temp_dept = the_person.get_destination().formal_name
    "You meet up in an empty office of the [temp_dept] department."
    $ mc.change_location(the_person.get_destination())
    $ scene_manager.add_actor(the_person, position="sitting", emotion="happy")
    the_person "Hello [the_person.mc_title], thank you for meeting me on such short notice."
    the_person "I have been thinking about how to make work more enjoyable for the staff."
    $ the_person.discover_opinion("work uniforms")
    if the_person.get_opinion_score("work uniforms") < 0:
        the_person "I understand why you want us to wear uniforms, I just hate being so restricted."
    elif the_person.get_opinion_score("work uniforms") > 0:
        the_person "I love having uniforms planned out for my work week, but sometimes they can get a bit stale."
    else:
        the_person "I don't feel too strongly about uniforms, but I know some of our staff does."
    $ the_person.discover_opinion("skimpy uniforms")
    if the_person.opinion.skimpy_uniforms < 0:
        the_person "I also think that sometimes they can be too revealing."
    elif the_person.opinion.skimpy_uniforms > 0:
        the_person "I also think they can provide an excellent excuse for showing off my body."
    else:
        the_person "I've also heard some mixed reviews on how skimpy they can be."
    mc.name "I understand where you're coming from, but I still think it's important to maintain uniform standards."
    the_person "I figured you would say that, but I don't want to get rid of them entirely."
    the_person "I was hoping that maybe I could help the girls pick out an outfit for next week and see if you like any of them enough to replace what we are using now."
    mc.name "Hmm, that is an interesting idea. Do you think they would be willing to let you use them as dress up dolls?"
    if get_existing_rivals(town_relationships, the_person):
        $ rival = get_random_from_list(get_existing_rivals(town_relationships, the_person))
        the_person "I think so. Although [rival.fname] might give me a bit of trouble, she is such a bitch."
    elif get_existing_friends(town_relationships, the_person):
        $ friend = get_random_from_list(get_existing_friends(town_relationships, the_person))
        the_person "I think so. If I have any trouble I can always get [friend.fname] to help me out."
    else:
        the_person "I don't think I'll have any trouble with that."
    menu:
        "Design some uniforms":
            mc.name "Alright, let's give this a try. Go ahead and get with the girls this weekend so that they have something to wear on Monday."
            mc.name "We'll meet up later in the week and discuss how they look."
            the_person "Thanks [the_person.mc_title], you won't regret this!"
            $ the_person.change_stats(happiness = 5, obedience = 2)
            $ uniform_prep_action = Action("Uniform prep", uniform_prep_requirement, "uniform_prep_label")
            $ mc.business.add_mandatory_crisis(uniform_prep_action)
        "Not now":
            mc.name "I still don't think it is a good idea. Let's stick with what we have already."
            the_person "Yes sir."
            $ number = the_person.get_opinion_score("work uniforms")*4
            $ the_person.change_stats(happiness = number, obedience = 2)
    $ scene_manager.clear_scene()
    return

label uniform_prep_label():
    $ the_person = mc.business.event_triggers_dict.get("uniform_designer", None)
    $ full_staff_design(the_person)
    $ uniform_followup_action = Action("Uniform followup", uniform_followup_requirement, "uniform_followup_label")
    $ mc.business.add_mandatory_crisis(uniform_followup_action)
    return

label uniform_followup_label():
    $ outfit = None
    $ stripped = True
    $ scene_manager = Scene()
    $ new_uniforms = 0
    $ all_uniforms = 0
    $ the_person = mc.business.event_triggers_dict.get("uniform_designer", None)
    $ forced_uniform_list = build_forced_uniform_list(the_person)
    $ uniform_followup_prep()
    "It has been a few days and you have to admit at least the girl's uniforms do have a bit more variety."
    "You take a moment out of your work day and track down [the_person.title] and talk to her about them."
    $ mc.change_location(the_person.location)
    $ scene_manager.add_actor(the_person, position = "sitting", display_transform = character_left_flipped)
    the_person "Hello, [the_person.mc_title]! How can I help you?"
    mc.name "I wanted to talk to you about the uniforms you picked out for the girls this week."
    the_person "Of course, I've been expecting this. We should probably get someone to come model their outfit so you can give me specific feedback."
    the_person "Was there a specific outfit that you like more than the rest?"
    $ done = False
    while not done:
        call screen main_choice_display(build_menu_items([get_sorted_people_list(forced_uniform_list, "Uniform review", "Back")]))
        $ person_choice = _return
        if person_choice != "Back":
            $ outfit = person_choice.event_triggers_dict.get("forced_uniform", None).get_copy()
            if "Uniform review" in forced_uniform_list:
                $ forced_uniform_list.remove("Uniform review")
            if "Back" in forced_uniform_list:
                $ forced_uniform_list.remove("Back")
            if person_choice in forced_uniform_list:
                $ forced_uniform_list.remove(person_choice)
            $ over = outfit.get_copy()
            $ under = outfit.get_copy()
            python:
                for clothing in over.get_full_strip_list(strip_feet = True):
                    if clothing.layer < 2:
                        over.remove_clothing(clothing)
                for clothing in under.get_full_strip_list(strip_feet = True):
                    if clothing.layer > 1:
                        under.remove_clothing(clothing)
                over.update_name()
                under.update_name()
            mc.name "I think the clothes [person_choice.fname] has been wearing look great."
            "You send a message to [person_choice.title] and ask her to come meet with you and [the_person.title]."
            "After some time you get a response..."
            person_choice "Be right there [person_choice.mc_title]!"
            $ scene_manager.update_actor(the_person, display_transform = character_left_flipped)
            $ scene_manager.add_actor(person_choice, outfit)
            $ scene_manager.apply_outfit(person_choice, outfit)
            "After a short time, [person_choice.title] makes her appearance, wearing the outfit that [the_person.title] helped design."
            mc.name "What do you think of the outfit that [the_person.fname] helped you pick out to wear this week?"
            if person_choice.judge_outfit(outfit, as_underwear = False, as_overwear = False):
                person_choice "I really like it."
                mc.name "That's good to hear, how would you feel about adding it to your normal uniform rotation?"
                person_choice "That would be great."
            else:
                person_choice "It is a bit revealing."
                mc.name "I'm sorry you feel that way. Do you think you might wear it if it was included in your normal uniform rotation?"
                person_choice "I don't know... I guess it would depend on what my other options were."
                if person_choice.judge_outfit(outfit, 20, as_underwear = False, as_overwear = False):
                    the_person "It isn't the worst outfit ever, so I might wear it occasionally."
                else:
                    the_person "It really makes me uncomfortable, but if there isn't anything better I won't have a choice."
            "Ultimately the decision is yours, what do you want to do?"
            menu:
                "Add to uniforms":
                    mc.name "Let's keep this, I think having it as an option will be a good thing."
                    the_person "Great, I'm so happy I could contribute to the company uniforms."
                    $ new_uniforms +=1
                    $ mc.business.business_uniforms.append(UniformOutfit(outfit))
                    $ mc.business.listener_system.fire_event("add_uniform", the_outfit = outfit)
                    $ mc.designed_wardrobe.add_outfit(outfit)
                "Discard design":
                    mc.name "It's just not what I picture when I think about our company."
                    the_person "Okay, I understand."
            if reduced_coverage_uniform_policy.is_active:
                $ stripped = True
                $ slut_continue_requirement = 0
                mc.name "Now, strip down to your underwear."
                if new_uniforms > 0:
                    mc.name "Let's see if the parts of the uniform are worth using as well."
                else:
                    mc.name "Maybe we can salvage something from this experiment."
                if not (person_choice.wearing_bra and person_choice.wearing_panties):
                    $ slut_continue_requirement = 40
                    person_choice "I... I can't do that [person_choice.mc_title]."
                    mc.name "What do you mean you can't? These are the rules you agreed with to work here, if you..."
                    "She shakes her head and interrupts you."
                    if not (person_choice.wearing_bra or person_choice.wearing_panties):
                        person_choice "No, I mean I can't strip down to my underwear because... I'm not wearing any."
                        $ slut_continue_requirement = 60
                        $ slut_continue_requirement += -(5*person_choice.opinion.not_wearing_anything)
                    elif not person_choice.wearing_bra:
                        person_choice "No, I mean I can't strip down to my underwear because... I'm not wearing a bra."
                        person_choice "My... breasts would be out."
                        $ slut_continue_requirement += -(5*person_choice.opinion.showing_her_tits)
                    else: #not wearing panties
                        person_choice "No, I mean I can't strip down to my underwear because... I'm not wearing any panties."
                        $ slut_continue_requirement += -(5*person_choice.opinion.showing_her_ass)
                    "You consider this for a moment, then shrug."
                    if new_uniforms > 0:
                        mc.name "That's unfortunate, I didn't realise she had designed an outfit without some items."
                    else:
                        mc.name "That's surprising, maybe I judged this outfit too soon."
                    mc.name "We better take a look."
                    person_choice "So you still want me to..."
                    mc.name "Strip. Now."
                    if person_choice.effective_sluttiness() < slut_continue_requirement:
                        "[person_choice.possessive_title!c] shuffles nervously."
                        person_choice "I don't think I can do it. I'm sorry [person_choice.mc_title]."
                        if office_punishment.is_active:
                            mc.name "If you're refusing the only choice I have is to write you up for disobedience, which will carry a penalty."
                            person_choice "I'm sorry, but I just can't do it. I'll accept a punishment if I have to."
                            $ person_choice.add_infraction(Infraction.disobedience_factory())
                            $ stripped = False
                        $ person_choice.change_happiness(-5)
                        $ person_choice.change_obedience(-1)
                        mc.name "Fine, we'll do it your way."
                    else:
                        $ person_choice.change_happiness(-5)
                        $ person_choice.change_obedience(2 + person_choice.opinion.being_submissive)
                        "Your words seem to shock her into action."
                elif person_choice.obedience < (130 - (10*person_choice.opinion.being_submissive)):
                    if person_choice.effective_sluttiness("underwear_nudity") < 40:
                        $ scene_manager.update_actor(person_choice, emotion = "angry")
                        person_choice "You... You can't make me do that!"
                        mc.name "Of course I can, it's company policy."
                        person_choice "But... It's not..."
                        mc.name "Strip. Now."
                        "[person_choice.possessive_title!c] glares at you, and for a moment you think she is going to refuse."
                        person_choice "Fine."
                        $ person_choice.change_obedience(1 + person_choice.opinion.being_submissive)
                    else:
                        person_choice "Okay."
                else:
                    if person_choice.effective_sluttiness("underwear_nudity") < 40:
                        "She blushes and looks away."
                        person_choice "Are you sure [person_choice.mc_title]? I'm not used to... being undressed in front of people."
                        mc.name "Of course I'm sure. Now strip."
                        "She nods meekly."
                        person_choice "Okay, if you say I have to..."
                    else:
                        "[person_choice.title] nods obediently."
                        person_choice "Yes [person_choice.mc_title], I'll go change."
                        mc.name "You'll change right here. Now strip."
                        person_choice "Of course. Right away."
                if person_choice.effective_sluttiness() > slut_continue_requirement:
                    $ generalised_strip_description(person_choice, person_choice.outfit.get_underwear_strip_list(visible_enough = False, avoid_nudity = False, strip_shoes = True))
                    $ scene_manager.update_actor(person_choice)
                    $ mc.change_locked_clarity(20)
                    if person_choice.update_outfit_taboos() or person_choice.effective_sluttiness() < 40:
                        "[person_choice.possessive_title!c] blushes and tries to cover her body."
                        person_choice "This is so embarrassing..."
                    $ slut_change = 0
                    if person_choice.tits_visible:
                        $ slut_change += person_choice.opinion.showing_her_tits
                    if person_choice.vagina_visible:
                        $ slut_change += person_choice.opinion.showing_her_ass
                    if person_choice.tits_visible and person_choice.vagina_visible:
                        $ slut_change += person_choice.opinion.not_wearing_anything
                    $ person_choice.change_slut(slut_change, 40)
                    mc.name "What do you think of the underwear that [the_person.title] helped you pick out to wear this week?"
                    if person_choice.judge_outfit(under, as_underwear = True, as_overwear = False):
                        person_choice "I really like it."
                        mc.name "That's good to hear, how would you feel about adding it to your normal uniform rotation?"
                        person_choice "That would be great."
                    else:
                        person_choice "It is a bit revealing."
                        mc.name "I'm sorry you feel that way. Do you think you might wear it if it was included in your normal uniform rotation?"
                        person_choice "I don't know... I guess it would depend on what my other options were."
                        if person_choice.judge_outfit(under, 20, as_underwear = True, as_overwear = False):
                            person_choice "It isn't the worst underwear ever, so I might wear it occasionally."
                        else:
                            person_choice "It really makes me uncomfortable, but if there isn't anything else better I won't have a choice."
                    "Ultimately the decision is yours, what do you want to do?"
                    menu:
                        "Add to uniforms":
                            mc.name "Let's keep this, I think having it as an option will be a good thing."
                            the_person "Great, I'm so happy I could contribute to the company uniforms."
                            $ new_uniforms +=1
                            $ mc.business.business_uniforms.append(UniformOutfit(under))
                            $ mc.business.listener_system.fire_event("add_uniform", the_outfit = under)
                            $ mc.designed_wardrobe.add_underwear_set(under)
                        "Discard design":
                            mc.name "It's just not what I picture when I think about our company."
                            the_person "Okay, I understand."
                    if minimal_coverage_uniform_policy.is_active:
                        mc.name "Great! Now to make sure we've covered all our bases lets see the overwear without underwear."
                        if not (person_choice.tits_visible and person_choice.vagina_visible and person_choice.tits_visible and person_choice.vagina_visible): #Something to strip
                            mc.name "Let's start by having you strip down."
                            $ generalised_strip_description(person_choice, person_choice.outfit.get_full_strip_list(strip_feet = True))
                            $ mc.change_locked_clarity(15)
                            if person_choice.update_outfit_taboos(): # Broke a taboo
                                "[person_choice.title] stands meekly in front of you, completely nude. She tries to cover herself up with her hands."
                        else:
                            mc.name "You're already undressed for the occasion."
                        mc.name "Go ahead and put on the overwear."
                        $ scene_manager.apply_outfit(person_choice, over)
                        $ scene_manager.update_actor(person_choice)
                        mc.name "What do you think of the overwear that [person_choice.title] helped you pick out to wear this week?"
                        if person_choice.judge_outfit(under, as_underwear = False, as_overwear = True):
                            person_choice "I really like it."
                            mc.name "That's good to hear, how would you feel about adding it to your normal uniform rotation?"
                            person_choice "That would be great."
                        else:
                            person_choice "It is a bit revealing, I guess I can try and find underwear that works with it."
                            mc.name "I'm sorry you feel that way. Do you think you might wear it if it was included in your normal uniform rotation?"
                            person_choice "I don't know... I guess it would depend on what my other options were."
                            if person_choice.judge_outfit(under, 20, as_underwear = True, as_overwear = False):
                                person_choice "It isn't the worst clothing ever, so I might wear it occasionally."
                            else:
                                person_choice "It really makes me uncomfortable, but if there isn't anything else better I won't have a choice."
                        "Ultimately the decision is yours, what do you want to do?"
                        menu:
                            "Add to uniforms":
                                mc.name "Let's keep this, I think having it as an option will be a good thing."
                                the_person "Great, I'm so happy I could contribute to the company uniforms."
                                $ new_uniforms +=1
                                $ mc.business.business_uniforms.append(UniformOutfit(over))
                                $ mc.business.listener_system.fire_event("add_uniform", the_outfit = over)
                                $ mc.designed_wardrobe.add_overwear_set(over)
                            "Discard design":
                                mc.name "It's just not what I picture when I think about our company."
                                the_person "Okay, I understand."
            $ all_uniforms += new_uniforms
            $ new_uniforms = 0
            menu:
                "Check another" if len(forced_uniform_list) > 0:
                    call uniform_dismissal_label(the_person, person_choice, stripped) from _call_uniform_dismissal_label_1
                "Finish up":
                    $ done = True
        else:
            $ done = True
            mc.name "Actually I don't think I like any of the outfits."
            the_person "Oh... okay. I guess we'll just go back to the uniforms you pick out."
    if all_uniforms > 1:
        the_person "Great, I'm so happy that you found something new worth using for our uniforms."
        the_person "As soon as you get a chance be sure to let us know which departments are going to be able to wear the new outfits."
        $ the_person.change_stats(happiness = all_uniforms, obedience = all_uniforms)
    elif all_uniforms > 0:
        the_person "Great, I'm so happy that you found something new worth using for our uniforms."
        the_person "As soon as you get a chance be sure to let us know which departments are going to be able to wear the new outfit."
    else:
        the_person "I'm sorry I couldn't find anything to your liking."
        $ the_person.change_stats(happiness = -5, obedience = 5)
    if renpy.random.randint(0,2) < all_uniforms:
        $ the_person.strengthen_opinion("work uniforms")
    the_person "Would you like the girls to go back to wearing their normal uniforms for the rest of the week?"
    menu:
        "Back to normal":
            mc.name "Yes, I really prefer what I picked out for them."
            $ full_staff_clear()
        "Keep wearing new uniforms" if all_uniforms > 0:
            mc.name "No, it has been a great change of pace seeing them in something different."
    the_person "What about for next week? Can I design some new uniforms for the girls?"
    menu:
        "Design more uniforms":
            mc.name "That would be great."
            $ uniform_prep_action = Action("Uniform prep", uniform_prep_requirement, "uniform_prep_label")
            $ mc.business.add_mandatory_crisis(uniform_prep_action)
        "Back to normal":
            mc.name "I'd rather that you didn't."
            $ mc.business.event_triggers_dict["uniform_designer"] = None
    if person_choice != "Back":
        call uniform_dismissal_label(the_person, person_choice, stripped) from _call_uniform_dismissal_label_2
    $ scene_manager.clear_scene()
    return

label uniform_dismissal_label(the_person, person_choice, stripped):
    mc.name "Thank you for coming to model for us [person_choice.title], you can head back to work."
    person_choice "Okay, thanks."
    if reduced_coverage_uniform_policy.is_active and stripped:
        "Since she has been dismissed, [person_choice.title] reaches for her discarded clothing, and you get a sudden idea."
        if minimal_coverage_uniform_policy.is_active:
            menu:
                "Keep her underwear":
                    mc.name "Actually, I think I like your outfit best the way it is now. You can stay like that for the rest of the week."
                    "She hesitates for a moment, but it isn't like that is an unreasonable request."
                    if person_choice.judge_outfit(over, as_underwear = False, as_overwear = False):
                        person_choice "Sure, I kinda like it better this way too."
                    else:
                        person_choice "...okay."
                        "It seems like she wants to argue but there is really no way for her to justify refusing to comply with this uniform."
                    $ person_choice.event_triggers_dict["forced_uniform"] = over.get_copy()
                "Take her overwear":
                    $ generalised_strip_description(person_choice, person_choice.outfit.get_underwear_strip_list())
                    $ scene_manager.apply_outfit(person_choice, under)
                    $ person_choice.event_triggers_dict["forced_uniform"] = under.get_copy()
                    "Once she has her underwear back on you stop her."
                    mc.name "Actually, leave those. I want to make sure that we get them listed in the uniform system properly, you can head back to work like you are."
                    if person_choice.judge_outfit(under, as_underwear = False, as_overwear = False):
                        person_choice "Sure, I kinda like it better this way too."
                    else:
                        if not under.wearing_bra and not under.wearing_panties:
                            person_choice "Like this? But I'm not really wearing anything."
                            mc.name "Exactly, I think I like you best this way."
                        elif not under.wearing_panties:
                            person_choice "Like this? But everyone is going to be able to see my pussy."
                            if not person_choice.has_taboo("vaginal_sex"):
                                mc.name "Exactly, and it will give me easier access if I get bored at work."
                            else:
                                mc.name "Exactly, and we all appreciate the show."
                        elif not under.wearing_bra:
                            person_choice "Like this? But everyone is going to be able to see my tits."
                            mc.name "Exactly, and we all appreciate the show."
                        else:
                            person_choice "...okay."
                            "It seems like she wants to argue but there is really no way for her to justify refusing to comply with this uniform."
                "Let her get dressed":
                    $ generalised_strip_description(person_choice, person_choice.outfit.get_underwear_strip_list())
                    $ scene_manager.apply_outfit(person_choice, outfit)
                    $ person_choice.outfit.restore_all_clothing()
        else:
            menu:
                "Keep her overwear":
                    mc.name "Actually, leave those. I want to make sure that we get them listed in the uniform system properly, you can head back to work like you are."
                    $ person_choice.event_triggers_dict["forced_uniform"] = under.get_copy()
                    if person_choice.judge_outfit(under, as_underwear = False, as_overwear = False):
                        person_choice "Sure, I kinda like it better this way too."
                    else:
                        if not under.wearing_bra and not under.wearing_panties:
                            person_choice "Like this? But I'm not really wearing anything."
                            mc.name "Exactly, I think I like you best this way."
                        elif not under.wearing_panties:
                            person_choice "Like this? But everyone is going to be able to see my pussy."
                            if not person_choice.has_taboo("vaginal_sex"):
                                mc.name "Exactly, and it will give me easier access if I get bored at work."
                            else:
                                mc.name "Exactly, and we all appreciate the show."
                        elif not under.wearing_bra:
                            person_choice "Like this? But everyone is going to be able to see my tits."
                            mc.name "Exactly, and we all appreciate the show."
                        else:
                            person_choice "...okay."
                            "It seems like she wants to argue but there is really no way for her to justify refusing to comply with this uniform."
                "Let her get dressed":
                    $ scene_manager.apply_outfit(person_choice, outfit)
                    $ person_choice.outfit.restore_all_clothing()
    $ person_choice.wear_uniform()
    $ scene_manager.remove_actor(person_choice)
    return

license: unknown


code:
init -1 python:
    def lily_intro_one_requirement(person): #friend
        if not lily.has_job(sister_student_job):
            return False
        if not person.event_triggers_dict.get("classmate_1", False):
            return person.location == university

    def lily_intro_two_requirement(person): #rival
        if not lily.has_job(sister_student_job):
            return False
        if person.event_triggers_dict.get("classmate_2", 999) + TIER_1_TIME_DELAY < day:
            return person.location == university

    def lily_intro_three_requirement(person): #acquaintance
        if not lily.has_job(sister_student_job):
            return False
        if person.event_triggers_dict.get("classmate_3", 999) + TIER_1_TIME_DELAY < day:
            return person.location == university

    lily_intro_one_action = Action("Meet Lily's friend", lily_intro_one_requirement, "lily_intro_one", menu_tooltip = "Meet one of Lily's classmates.")

    def college_girls_mod_initialization():
        lily.add_unique_on_room_enter_event(lily_intro_one_action)

    college_girl_test_action = Action("College Girl Test", kaden_test_req, "college_girl_test", menu_tooltip = "Meet three college girls.")

init 3 python:
    def lily_friend_mod_initialization(): #Add actionmod as argument#
        global lily_friend
        lily_friend = make_person(age = lily.age, pubes_style = default_pubes, personality = introvert_personality, \
            stat_array = [3,2,4], skill_array = [1,1,1,1,1], sex_skill_array = [1,1,1,1], \
            sluttiness = 0, obedience_range = [105, 110], happiness = 119, love = 10, start_home = None, \
            mc_title = mc.name, relationship = "Single", kids = 0, \
            forced_opinions = [["flirting", 2, False], ["skirts", 2, False], ["the colour blue", 2, False]], \
            forced_sexy_opinions = [["getting head", 2, False], ["drinking cum", -2, False], ["giving blowjobs", -2, False], ["public sex", -2, False]])
        lily_friend.generate_home().add_person(lily_friend)
        lily_friend.home.background_name = "student_apartment_background"
        university_home_hub.people.append(lily_friend)
        lily_friend.set_title(lily_friend.name)
        lily_friend.set_possessive_title(lily_friend.name)
        lily_friend.change_job(student_job)
        university.add_person(lily_friend)
        # set relationships
        town_relationships.update_relationship(lily, lily_friend, "Friend")
        return

    def lily_rival_mod_initialization(): #Add actionmod as argument#
        global lily_rival
        lily_rival = make_person(age = lily.age, personality = wild_personality, pubes_style = shaved_pubes, \
            stat_array = [2,1,3], skill_array = [1,1,1,1,1], sex_skill_array = [5,2,2,2], \
            sluttiness = 35, obedience_range = [115, 120], happiness = 110, love = 0, \
            mc_title = mc.name, relationship = "Single", kids = 0, \
            forced_opinions = [["flirting", 2, False], ["the colour green", 2, False], ["skirts", 2, False]], \
            forced_sexy_opinions = [["taking control", 2, False], ["getting head", 2, False], ["drinking cum", 2, False], ["giving blowjobs", 2, False], ["public sex", 2, False]])
        if not lily_rival.has_large_tits:
            lily_rival.tits="DD"
        lily_rival.generate_home().add_person(lily_rival)
        lily_rival.home.background_name = "student_apartment_background"
        university_home_hub.people.append(lily_rival)
        lily_rival.set_title(lily_rival.name)
        lily_rival.set_possessive_title(lily_rival.name)
        lily_rival.change_job(student_job)
        university.add_person(lily_rival)
        # set relationships
        town_relationships.update_relationship(lily_friend, lily_rival, "Friend")
        town_relationships.update_relationship(lily, lily_rival, "Rival")
        return

    def lily_neutral_mod_initialization(): #Add actionmod as argument#
        global lily_neutral
        lily_neutral = make_person(age = lily.age, personality = relaxed_personality, pubes_style = landing_strip_pubes, \
            stat_array = [3,2,4], skill_array = [1,1,1,1,1], sex_skill_array = [2,2,2,2], \
            sluttiness = 15, obedience_range = [95, 100], happiness = 119, love = 0, \
            mc_title = mc.name, relationship = "Single", kids = 0, \
            forced_opinions = [["flirting", 0, False], ["the colour red", 2, False], ["skirts", 2, False]], \
            forced_sexy_opinions = [["getting head", 2, False], ["drinking cum", 0, False], ["giving blowjobs", 0, False], ["public sex", 0, False]])
        lily_neutral.generate_home().add_person(lily_neutral)
        lily_neutral.home.background_name = "student_apartment_background"
        university_home_hub.people.append(lily_neutral)
        lily_neutral.set_title(lily_neutral.name)
        lily_neutral.set_possessive_title(lily_neutral.name)
        lily_neutral.change_job(student_job)
        university.add_person(lily_neutral)
        # set relationships
        town_relationships.update_relationship(lily_friend, lily_neutral, "Friend")
        town_relationships.update_relationship(lily_rival, lily_neutral, "Acquaintance")
        town_relationships.update_relationship(lily, lily_neutral, "Acquaintance")
        return

init 16 python: # hijack
    add_label_hijack("normal_start", "activate_college_girls_mod_core")
    add_label_hijack("after_load", "update_college_girls_mod_core")

label activate_college_girls_mod_core(stack):
    python:
        college_girls_mod_initialization()
        execute_hijack_call(stack)
    return

label update_college_girls_mod_core(stack):
    python:
        if "lily_friend" not in globals():
            college_girls_mod_initialization()
        execute_hijack_call(stack)
    return

label lily_intro_one(the_sister): #friend
    $ scene_manager = Scene()
    if "lily_friend" not in globals():
        $ lily_friend_mod_initialization()
    $ lily.event_triggers_dict["classmate_1"] = True
    $ lily.event_triggers_dict["classmate_2"] = day
    $ the_sister = lily
    $ the_sister.apply_outfit(limited_university_wardrobe.decide_on_outfit(the_sister))
    $ the_person = lily_friend
    $ the_person.apply_outfit(limited_university_wardrobe.decide_on_outfit(the_person))
    $ the_person.set_mc_title(mc.name)
    $ scene_manager.add_actor(the_sister, the_sister.outfit, emotion = "happy")
    $ scene_manager.add_actor(the_person, the_person.outfit, display_transform = character_left_flipped(xoffset = 0.45), position = "walking_away", emotion = "happy")
    "As you make your way across campus you spot [the_sister.title] standing and talking to another student."
    "She spots you and waves you over to join her."
    the_sister "Hey, [mc.name], what are you doing on campus?"
    menu:
        "Visiting Nora":
            $ the_sister.change_stats(obedience = 2, happiness = -2)
            mc.name "I was on my way to see Professor [nora.title], she has been helping me with some of the research for my business."
            the_sister "That's cool, how is it being back on campus after graduation?"
            mc.name "A little weird, not having a class to get to or a project to work on makes it feel oddly relaxed."
        "Visiting you":
            $ the_sister.change_stats(happiness = 5, love = 2)
            mc.name "We've both been so busy I figured I would come say hi while I was running some errands."
            the_sister "That's sweet, I've been busy too, never seems to be enough time to get my class work all done."
            mc.name "Yeah, I remember what it was like trying to balance classes and homework while trying to have a social life."
    $ scene_manager.update_actor(the_person, display_transform = character_left_flipped(xoffset = 0.375), position = "stand5")
    the_person "Erm, hello..."
    the_sister "Oh, I'm so sorry, [mc.name] this is my friend [the_person.fname]. [the_person.fname], this is my brother [mc.name]."
    mc.name "It's a pleasure to meet you."
    the_person "You too. [the_sister.fname] we should be getting to class."
    "[the_sister.title] looks at her phone to check the time."
    the_sister "Yeah, you're right, sorry [mc.name] I'll see you at home tonight."
    mc.name "Alright, have fun in class. It was nice meeting you [the_person.fname]."
    $ scene_manager.update_actor(the_sister, position = "walking_away")
    the_person "Yeah, you too."
    $ scene_manager.update_actor(the_person, position = "walking_away")
    "They both turn to walk away, and you can't help but to stay and watch until they clear your line of sight."
    $ scene_manager.update_actor(the_person, display_transform = character_left_flipped(xoffset = .35, zoom = 0.8))
    $ scene_manager.update_actor(the_sister, display_transform = character_right(xoffset = -.05, zoom = 0.8))
    "Your connection to [nora.fname] means that you are spending more time on campus than you expected when you graduated."
    $ scene_manager.update_actor(the_person, display_transform = character_left_flipped(xoffset = .325, zoom = 0.6))
    $ scene_manager.update_actor(the_sister, display_transform = character_right(xoffset = -.1, zoom = 0.6))
    "It occurs to you that there is some pretty strong potential for you to get more involved with the student body as an alumnus, not to mention the student's bodies."
    $ scene_manager.update_actor(the_person, display_transform = character_left_flipped(xoffset = .3, zoom = 0.4))
    $ scene_manager.update_actor(the_sister, display_transform = character_right(xoffset = -.15, zoom = 0.4))
    "[mc.business.name] is a pretty big time commitment, but as you get things under control it might be worth seeing how you could get involved."
    $ scene_manager.update_actor(the_person, display_transform = character_left_flipped(xoffset = .275, zoom = 0.2))
    $ scene_manager.update_actor(the_sister, display_transform = character_right(xoffset = -.2, zoom = 0.2))
    "With the right connections you are sure that it could be a beneficial relationship for you to grow."
    $ scene_manager.clear_scene()
    $ scene_manager = None
    python:
        lily_intro_two_action = Action("Meet Lily's rival", lily_intro_two_requirement, "lily_intro_two", menu_tooltip = "Meet one of Lily's classmates.")
        lily.add_unique_on_room_enter_event(lily_intro_two_action)
    return

label lily_intro_two(the_sister): #rival
    $ scene_manager = Scene()
    if "lily_rival" not in globals():
        $ lily_rival_mod_initialization()
    $ lily.event_triggers_dict["classmate_2"] = 999
    $ lily.event_triggers_dict["classmate_3"] = day
    $ the_sister = lily
    $ the_person = lily_rival
    $ the_sister.apply_outfit(limited_university_wardrobe.decide_on_outfit(the_sister))
    $ the_person.apply_outfit(limited_university_wardrobe.decide_on_outfit(the_person))
    $ the_person.set_mc_title(mc.name)
    $ followup = None
    $ scene_manager.add_actor(the_sister, the_sister.outfit, position = "walking_away", emotion = "happy")
    "As you make your way across campus you are enjoying the view of the undergrads wandering around in their uniforms."
    "One of them looks more familiar than the rest, and you make your way up behind [the_sister.possessive_title]."
    "You reach out and tap her on the right shoulder as you step to the left, ending up behind her as she turns."
    $ scene_manager.update_actor(the_sister, position = "stand3")
    the_sister "Hey... um... oh, there you are. Hey [the_person.mc_title]!"
    mc.name "Hey [the_sister.title], how are you doing today?"
    if the_sister.sluttiness > 40:
        the_sister "I'm alright, wishing I could be at home playing around on InstaPic instead of here worrying about class."
    elif the_sister.love > 40:
        the_sister "I'm fine, although I'd be happier if we could be hanging out at home together."
    elif the_sister.happiness < 100:
        the_sister "Honestly, I'm having a rough time. I just can't seem to get everything together."
    else:
        the_sister "Eh, you know school is school, but I'm fine."
    $ scene_manager.add_actor(the_person, the_person.outfit, position = "stand4", display_transform = character_right(xoffset = -0.2), emotion = "happy")
    "The two of you chat for a bit when suddenly another student steps up interjecting herself in your conversation."
    the_person "Hello there handsome, I haven't seen you around here before, planning to transfer?"
    the_sister "Excuse us, we were talking."
    the_person "That's why I stepped in, can't have you making a bad impression with one of the visitors to our campus."
    $ scene_manager.update_actor(the_person, position = "stand2", display_transform = character_right(xoffset = -0.05))
    "She steps forward even more, turning her shoulder to slightly block [the_sister.title] and smiling at you."
    the_person "Hello, I'm [the_person.fname], one person welcoming committee, is there anything I can do to help you?"
    menu:
        "Stand up for [the_sister.fname]":
            $ the_sister.change_stats(love = 2, happiness = 2)
            $ the_person.change_stats(happiness = -2, obedience = -2)
            mc.name "Actually no, I'm quite happy talking with [the_sister.fname]. Why don't you go find someone else to bother?"
            the_person "Really, you'd rather spend time with her than with someone like me?"
            "She leans forward squeezing her arms together to draw further attention to her rather sizable chest."
            menu:
                "Dismiss [the_person.fname]":
                    mc.name "Wow, really just whoring it up out here. Tell me, would you expect me to pay for it after?"
                    $ scene_manager.update_actor(the_person, emotion = "angry")
                    the_sister "Don't worry, [the_sister.mc_title] she gives it away all over campus."
                    the_person "Whatever, you too have fun being prudes. [the_person.mc_title] give me a call if you ever want to have some fun, you're cute enough to get a second chance."
                    $ followup = lily
                "Give in":
                    mc.name "Well that is certainly and interesting offer. [the_sister.fname] we can catch up later tonight."
                    $ scene_manager.update_actor(the_sister, emotion = "sad")
                    the_sister "Fine, I'll see you at home. Just be sure to use protection with this one."
                    the_person "At least he won't need to put a bag over my head."
                    $ followup = lily_rival
        "Greet [the_person.fname]":
            $ the_sister.change_stats(love = -2, happiness = -2)
            $ the_person.change_stats(happiness = 2, obedience = 2)
            mc.name "Well, hello, I'm [mc.name] and I certainly wouldn't mind getting your number if you are offering."
            the_person "Of course, let me put it in your phone for you."
            "She reaches out and accepts your phone to start typing in her number."
            the_person "You can go [the_sister.fname], we won't be needing you for anything."
            $ scene_manager.update_actor(the_sister, emotion = "sad")
            "[the_sister.fname] flashes [the_person.fname] a dirty look and then looks to you, disappointment plain on her face."
            the_sister "Well, I guess I'll see you at home later tonight [the_sister.mc_title]."
            $ mc.phone.register_number(the_person)
            "[the_person.fname] hands back your phone, a sudden realisation flashing across her face."
            the_person "Oh, you're family, no wonder you are wasting time with her. Can't help who you are related to."
            menu:
                "Defend [the_sister.fname]":
                    mc.name "Hey now, you may have a terrible family but I happen to love mine."
                    $ scene_manager.update_actor(the_person, emotion = "angry")
                    the_sister "Yeah, [the_person.fname], stop taking out your broken home syndrome on the rest of us."
                    the_person "Whatever, give me a call if you ever want to have some fun [the_person.mc_title], you're cute enough to get a second chance."
                    $ followup = lily
                "Let it go":
                    mc.name "Yeah... although [the_sister.fname] isn't so bad."
                    the_person "Sure, sure, gotta keep up appearances. Let's go somewhere private where you can be a bit more honest with me."
                    the_sister "Fine, I guess I know when I'm not wanted."
                    $ followup = lily_rival
    if followup == lily:
        $ the_sister.change_stats(love = 2, happiness = 2)
        $ the_person.change_stats(happiness = -2, obedience = -2)
        $ scene_manager.update_actor(the_person, position = "walking_away", display_transform = character_center)
        "[the_person.title] turns to walk away, aggressively swinging her hips trying to draw more attention to herself."
        the_sister "I'm sorry about that, [the_person.fname] has been a pain in the ass since the semester started."
        $ scene_manager.remove_actor(the_person)
        mc.name "I can see that, sorry you have to deal with that on top of the normal hassles of classes."
        the_sister "Oh, class, I have to go or I'm going to be late. See you tonight."
        "[the_sister.possessive_title!c] turns and starts to jog off in the direction of her next class."
    else:
        $ the_sister.change_stats(love = -2, happiness = -2)
        $ the_person.change_stats(happiness = 2, obedience = 2)
        $ scene_manager.update_actor(the_sister, position = "walking_away")
        "[the_sister.possessive_title!c] turns and walks away, clearly upset by the way you treated her."
        the_person "You made the right decision, let's go and I'll reward you for your display of intelligence."
        $ scene_manager.remove_actor(the_sister)
        "[the_person.title] wraps herself around your arm and starts to pull you towards one of the nearby buildings."
        the_person "So... [the_sister.fname] is she super annoying at home because she sure is in class?"
        mc.name "She can be, it really depends on the day."
        the_person "You poor thing, come in here and I'll make you feel better."
        "She pulls you into an empty meeting room and locks the door."
        the_person "So, tell me, how can I reward you?"
        $ the_person.break_taboo("sucking_cock")
        call get_fucked(the_person, the_goal = "get mc off", private= True, start_position = blowjob, skip_intro = False, ignore_taboo = True, allow_continue = True) from _call_get_fucked_lily_rival
        #"Did that feel good? I just want to make you feel good..."
        $ scene_manager.update_actor(the_person, position = "kneeling1")
        menu:
            "Be grateful":
                mc.name "You certianly accomplished that. I'm real glad I came to campus today."
                $ scene_manager.update_actor(the_person, position = "stand2")
                "You help [the_person.title] to her feet and watch as she adjusts her clothes."
                $ scene_manager.review_outfits()
                "You tuck your cock away and zip up your pants while she is busy."
                "Once you're both presentable you resume your conversation."
                the_person "Well, I'm here five days a week if you want to come back. Just don't bring your sister."
                mc.name "I'll keep that in mind."
                $ the_person.change_stats(love = 2)
            "Be dismissive":
                mc.name "Eh, you did okay, but I've had better."
                "You stare down at her as you adjust your pants and continue."
                mc.name "Maybe I'll come back and let you practice more sometime."
                "She seems a bit stunned, but you don't wait for an answer before turning to walk away."
                $ the_person.change_stats(love = -2)
                $ scene_manager.remove_actor(the_person)
        "You head back out to campus to resume your day."
        "It was a fun distraction, but you feel a bit bad about the way you treated [the_sister.possessive_title]."
    $ scene_manager.clear_scene()
    $ scene_manager = None
    python:
        lily_intro_three_action = Action("Meet Lily's acquaintance", lily_intro_three_requirement, "lily_intro_three",
            menu_tooltip = "Meet one of Lily's classmates.")
        lily.add_unique_on_room_enter_event(lily_intro_three_action)
    return

label lily_intro_three(the_sister): #acquaintance
    $ scene_manager = Scene()
    if "lily_neutral" not in globals():
        $ lily_neutral_mod_initialization()
    $ lily.event_triggers_dict["classmate_3"] = 999
    $ the_sister = lily
    $ the_person = lily_neutral
    $ the_sister.apply_outfit(limited_university_wardrobe.decide_on_outfit(the_sister))
    $ the_person.apply_outfit(limited_university_wardrobe.decide_on_outfit(the_person))
    $ the_person.set_mc_title(mc.name)
    "Another day, and you once again find yourself on campus. When you graduated you never imagined you would spend so much time back here, but at least you get to see [the_sister.title] more often as a result."
    "You are fairly familiar with her schedule now and it doesn't take long to spot her making her way between buildings."
    $ scene_manager.add_actor(the_sister, the_sister.outfit, emotion = "happy")
    mc.name "Hey, [the_sister.title], how are you doing today?"
    the_sister "Pretty good, just making my way to my next class—what about you?"
    mc.name "I'm great, just stopping by to drop off some papers for [nora.fname] and figured we could chat for a bit while I was here."
    $ scene_manager.add_actor(the_person, the_person.outfit, display_transform = character_center, position = "stand4", emotion = "happy")
    "Suddenly, another girl steps up from behind you, slightly out of breath."
    the_person "Hey, [the_sister.fname], no time to flirt, we have class in two minutes."
    the_sister "I'm not flirting, this is my brother [mc.name]. [mc.name] this is [the_person.fname]"
    mc.name "It's a pleasure to meet you."
    the_person "Sorry, nice to meet you, but we really do have to go."
    "She grabs [the_sister.fname]'s hand and gently pulls her away."
    $ scene_manager.update_actor(the_person, display_transform = character_center(xoffset = 0.1), position = "walking_away")
    $ scene_manager.update_actor(the_sister, position = "back_peek")
    the_sister "Sorry, we really do have to go."
    mc.name "Alright, have a good time in class."
    $ scene_manager.update_actor(the_sister, position = "walking_away")
    "Once they are turned around they both break into a jog to try and make it to class in time."
    $ the_sister.set_side_job(study_buddy_job)
    $ starter_buddy = get_random_from_list([lily_friend, lily_rival, lily_neutral])
    $ starter_buddy.set_side_job(study_buddy_job)
    if not starter_buddy.is_favourite:
        $ starter_buddy.toggle_favourite()
    if day%7 == 0:
        call study_buddy_prep_label from _call_study_buddy_prep_label_first
    $ scene_manager.clear_scene()
    $ scene_manager = None
    return

label college_girl_test():
    $ university.show_background()
    "LILY INTRO ONE"
    call lily_intro_one(lily) from _call_lily_intro_one_test
    "LILY INTRO TWO"
    call lily_intro_two(lily) from _call_lily_intro_two_test
    "LIILY INTRO THREE"
    call lily_intro_three(lily) from _call_lily_intro_three_test
    $ mc.location.show_background()
    return